

LECCIÓN 5

En cualquier tipo de lenguaje de programación usaremos instrucciones y funciones para que la aplicación funcione. En esta lección vamos a conocer algunas de estas estructuras básicas para la programación. Estas estructuras pueden utilizarse solas o conjuntas de forma que formen un programa. Si las instrucciones básicas las tenemos claras desde un principio, tenemos mucho terreno ganado ya que en la gran parte de aplicaciones que programemos utilizaremos estas instrucciones combinadas con más o menos dificultad.

Estructuras básicas en la programación

Vamos a ver que muchos programas necesitan de unas estructuras para su buen funcionamiento. En los siguientes apartados vamos a ir viendo estas estructuras una a una, comentando su utilidad y su escritura. Veremos que una vez conozcamos su funcionamiento es muy fácil pensar en ellas cuando se nos plantea un nuevo problema. Por esta razón pensamos que esta lección es de suma importancia y te recomendamos le dediques todo el tiempo que puedas.

Estructuras de decisión

Las líneas de un programa, dentro de un evento, se ejecutan de arriba abajo. Pero en muchas ocasiones no nos interesa esta linealidad con lo que podemos cambiar el orden de las líneas de código según nos convenga y según la utilidad le queramos dar. Estas estructuras nos permiten tomar decisiones según las condiciones que se den en nuestra aplicación. Por ejemplo, podemos tener una aplicación en la que nos interese calcular el peso ideal de una persona dada, su estatura y su sexo. Pues bien, el sexo será la condición que marcará que camino hay que tomar, ya que si es un hombre tendremos que multiplicar su estatura por una valor y si es una mujer por otra.

Vamos a ir viendo las principales estructuras de decisión una a una.

If...Then...End If

La estructura básica de esta instrucción es la siguiente:

```
If (Condición) Then
  [Instrucciones Verdadero]
Else
  [Instrucciones Falso]
End If
```

Los corchetes muestran partes opcionales de la instrucción.

Condición: Aquí escribiremos la condición a evaluar, para que el ordenador nos devuelva una respuesta y según esta respuesta podamos actuar. Vamos a retomar el ejemplo que hemos planteado anteriormente. Nosotros podríamos tener un **TextBox** llamado sexo, en el que escribiremos una "V" para varón o una "H" para hembra. En el lugar reservado para la **condición** podremos escribir **Sexo.Text = "H"**, quedando la línea completa de la siguiente manera **If Sexo.Text = "H" Then**. Fíjate que aquí no estamos haciendo una asignación, si no que lo que hacemos es plantearnos una pregunta que podemos leer más o menos de la siguiente forma: **Si el contenido del objeto Sexo es H, entonces...** a lo que el ordenador nos devolverá **verdadero** o **falso**. Según la respuesta que nos retorna el ordenador nosotros podremos actuar de una forma u otra.

[Instrucciones Verdadero]: Aquí pondremos la instrucción o instrucciones que se deberá realizar si la respuesta a la condición es **Verdadera**.

Si nosotros solo tenemos una instrucción para colocar en el lugar de **[Instruc-**

ciones Verdadero] y ninguna en **[Instrucciones Falso]** podemos escribir esta estructura de la siguiente manera:

If [Condición] Then [Instrucción Verdadero]

Recuerda que en esta estructura sólo se puede escribir una instrucción en caso que la **condición** sea **verdadera**.

Si nosotros lo que queremos escribir son varias instrucciones, pero no queremos que el ordenador realice nada en el caso de que la condición sea **Falsa** podemos escribir la estructura de la siguiente forma:

If [Condición] Then [Instrucciones Verdadero] End If

Las instrucciones para realizar en caso que la condición sea **verdadera** se escribirán una debajo de la otra. Estas instrucciones están escritas entre la partícula **If** y **End If**.

. Práctica 1

Vamos a realizar una pequeña aplicación para que el ordenador nos diga, después de introducir la edad de una persona si es mayor de edad o no. Consideraremos la mayoría de edad a los 18 años.

En esta práctica simplificaremos los controles a utilizar, si lo deseas puedes ampliar la práctica tanto como desees.

1. Coloca, en el lugar que desees de un formulario nuevo, un **TextBox** y un **Label**.

El **TextBox** lo utilizaremos para introducir la edad y el **Label** para que el ordenador nos devuelva la cadena "**Mayor de edad**" en caso de ser más grande o igual a 18 años, o la cadena "**Menor de edad**" en caso de ser menor de 18 años.

2. Cambia la propiedad **(Nombre)** del **TextBox** y escribe: **Edad**.

3. Cambia la propiedad **(Nombre)** del **Label** y escribe: **Comentario**.

Puedes borrar el contenido de ambos objetos y modificar el aspecto como tú quieras.

4. Inserta un botón, en el cual colocaremos el código que se ejecutará al hacer clic sobre él.

5. Pon en la propiedad **Caption** de dicho botón **Calcular**. (No hace falta que cambies la propiedad **(Nombre)**).

6. Haz doble clic en el botón **Calcular**.

7. Escribe el siguiente código:

```
If Val(Edad.Text) < 18 Then  
Comentario.Caption = "Es menor de edad"  
Else  
Comentario.Caption = "Es mayor de edad"  
End If
```

Vamos a comentar el código anterior. El ordenador lo primero que hace es mirar

el valor del contenido del objeto llamado **Edad**. Este objeto es de tipo texto y nosotros lo que estamos haciendo es mirar si es mayor o menor que un número. Por esta razón nosotros convertimos el texto en valor numérico utilizando la orden **Val**. El ordenador se hace la pregunta "¿El contenido de **Edad** es menor que 18?. Si la respuesta es **verdadera**, pasa a la primera parte del **If** y escribe en el objeto **Comentario** la frase "Es menor de edad", si la respuesta es **falsa**, pasa a la segunda parte del **If**, donde se escribe "Es mayor de edad".

8. Haz una ejecución de prueba.

. Práctica 2

Vamos a utilizar este mismo ejemplo, para explicar como escribiríamos el código en caso que solo quisiésemos que el ordenador nos devolviese un mensaje en caso que la edad fuera menor de 18 años.

1. Accede al código del botón **Calcular**.

2. Borra el código que hemos escrito anteriormente y escribe el siguiente código:

```
If Val(Edad.Text) < 18 Then Comentario.Caption = "Es menor de edad"
```

Todo este código deberá estar escrito en la misma línea.

Observa que no hemos utilizado la parte **Else** de la estructura, ni instrucciones que van en la parte de **falso**, ni **End If** indicando el final del código. Esto es debido a que solo queremos realizar una operación en caso de que sea Verdadero, con lo que no hace falta "cerrar" el **If**, y no ponemos **Else** porque no queremos instrucciones en el caso que sea la respuesta falsa.

3. Realiza un ejecución de prueba.

Ten en cuenta que si cuando ejecutas el programa escribes 21, la aplicación nos devolverá un mensaje diciendo "Es menor de edad". Si acto seguido borras la edad y escribes 10, al pulsar el botón continuarás viendo en el cuadro de mensaje "Es menor de edad". Esto no quiere decir que la aplicación funcione mal, ya que no existe ninguna instrucción que haga que en el caso de no ser mayor de edad se borre el mensaje.

Podemos decir que la aplicación funciona correctamente, pero está mal diseñada, ya que nos proporciona información incorrecta.

. Práctica 3

Imagina que lo que queremos ahora es que el ordenador nos devuelva, mirando la edad que introducimos, si es menor de 10 años, si tiene entre 10 y 20 años, si tiene entre 20 y 30 años o si es mayor de 30. Hasta este momento solo hemos visto la instrucción **If** para poder controlar un caso como el que planteamos.

Veamos que código tendríamos que usar para que funcione el caso anteriormente planteado.

1. Escribe estás líneas dentro del botón **Calcular**. Borra las instrucciones escritas anteriormente.

```
1 If Val(Edad.Text) < 10 Then  
2     Comentario.Caption = "Menos de 10 años"  
3 Else
```

```

4      If Val(Edad.Text) < 20 Then
5          Comentario.Caption = "Entre 10 y 20 años"
6      Else
7          If Val(Edad.Text) < 30 Then
8              Comentario.Caption = "Entre 20 y 30
años"
9          Else
10             Comentario.Caption = "Más de 30
años"
11         End If
12     End If
13 End If

```

2. Realiza una ejecución de prueba y observa el funcionamiento introduciendo diferentes valores.

Vamos a plantear un caso hipotético para ver como funcionaría este código.

Imagina que el usuario tiene 25 años. El ordenador comenzaría en la línea **1**, donde le preguntamos si el usuario tiene menos de 10 años. Como la respuesta es falsa pasamos a la línea **4**. Aquí preguntamos si el usuario tiene menos de 20 años. En este caso también es falso, con lo que pasamos a la línea **7**. Hacemos otra pregunta mirando si el usuario tiene menos de 30 años. En este caso la respuesta es Verdadera con lo que pasamos a la línea **8** donde escribimos el mensaje correspondiente en el objeto **Comentario** después de esto pasamos a las líneas **11**, **12** y **13** donde termina el código del evento.

En el ejemplo anterior hemos visto como añadir un **If** dentro de otro, observa como cada **If** tiene su **Else** y su **End If** asociados. Observa las líneas **1**, **3** y **13**, las líneas **4**, **6** y **12**, y las líneas **7**, **9** y **11**. Habrás podido comprobar que hemos utilizado tabuladores en el momento de escribir el código, esto se hace para poder facilitar la lectura de las líneas de código, ya que de esta manera podemos ver con un golpe de vista, donde empieza y donde acaba un **If**.

Vamos a aprovechar este ejemplo para explicar otra estructura de decisión.

Select Case

Esta estructura la utilizaremos en los casos en los que tengamos muchas condiciones a evaluar, ya que con la estructura **If** se podría complicar bastante.

En esta nueva estructura de decisión se valoran los diferentes valores que puede tomar una determinada expresión y según el valor que tenga esta se actúa en consecuencia.

Es importante destacar que en esta instrucción no miramos si una pregunta o condición es verdadera o falsa, actuando así en consecuencia, sino que miramos el valor que toma un determinado objeto o variable.

```

Select Case [Expresión para comparar]
Case [Expresión 1]
    [Instrucciones 1]
...
Case [Expresión n]
    [Instrucciones n]
[Case Else]
    [Instrucciones Else]
End Select

```

En **[Expresión para comparar]** pondremos el objeto sobre el cual queremos

preguntar por su valor.

En **[Expresión 1]** escribiremos cual es la pregunta que deseamos hacer sobre el valor escrito anteriormente.

En **[Instrucciones 1]** pondremos las instrucciones que se realizarán en caso de que **[Expresión 1]** sea verdadera.

Podremos poner tantas **[Expresiones]** como queramos.

Si queremos que se haga algo en caso que ninguna de las expresiones que hemos puesto anteriormente se cumpla, podemos escribir **[Case Else]** y seguidamente la o las instrucciones que se tienen que realizar.

No olvides, igual que en el caso del **If**, cerrar la expresión utilizando **End Select**.

. Práctica 4

1. Borra el código del botón **Calcular**.
2. Escribe en su interior el siguiente código:

```
Select Case Val(Edad.Text)
    Case < 10
        Comentario.Caption = "Menos de 10 años"
    Case < 20
        Comentario.Caption = "Entre 10 y 20 años"
    Case < 30
        Comentario.Caption = "Entre 20 y 30 años"
    Case Else
        Comentario.Caption = "Más de 30 años"
End Select
```

3. Haz una ejecución de prueba.

Intenta comparar los dos últimos códigos que te hemos planteado y observa que el segundo es mucho más claro y fácil de entender que el primero en el que utilizábamos **If** anidados.

Operadores de comparación

En los ejemplos anteriores hemos podido ver como estabamos mirando en todo momento si la cantidad introducida era menor (<) a unos valores determinados. Como puedes suponer hay varios operadores que nos permitirán hacer diferentes comparaciones. Ha estos elementos le llamaremos **Operadores de comparación**.

Esta es la lista de **operadores de comparación** que podemos utilizar en **Visual Basic**.

=	Igual a
<>	Distinto
<	Menor
>	Mayor
<=	Menor o igual
>=	Mayor o igual

Estos operadores nos permiten hacer comparaciones entre diferentes elementos.

Vamos a realizar el mismo ejemplo que antes pero utilizando como operador de comparación Mayor (>).

```

Select Case Val(Edad.Text)
  Case > 30
    Comentario.Caption = "Más de 30 años"
  Case > 20
    Comentario.Caption = "Entre 20 y 30 años"
  Case > 10
    Comentario.Caption = "Entre 10 y 20 años"
  Case Else
    Comentario.Caption = "Menos de 10 años"
End Select

```

Observa las modificaciones realizadas en el código.

Operadores lógicos

Con los operadores lógicos podemos mirar condiciones que tengan de depender de más de un criterio. Como puede ser: Tener más de 18 años **y** medir 1'70 m.

Estos son los operadores lógicos principales: Y, O y No.

And	Y
Or	O
Not	No

Todos estos operadores lógicos se utilizan para hacer una pregunta entre dos elementos.

Vamos a verlos uno a uno.

And

La conjunción **And** nos permite hacer una pregunta entre dos elementos.

Para que el resultado total sea **verdadero** cada una de las partes de la pregunta tiene que ser **verdadero**.

Pongamos un ejemplo: imagina que tenemos una aplicación en la que debe aparecer un mensaje en caso de que una persona sea mayor de 25 años y esté casada. Para esto tendríamos que escribir la línea de código de la pregunta de la siguiente manera:

If Edad.Text > 25 And Casado.Value = True Then

(Los objetos de este ejemplo son imaginarios)

Hagamos una tabla en la que podemos ver con más claridad en que casos se mostraría el mensaje.

Edad.Text	Casado.Value	Mensaje
Verdadero	Verdadero	Sí
Verdadero	Falso	No
Falso	Verdadero	No
Falso	Falso	No

Observa que solo se mostrará el mensaje en el momento en el que las dos

partes de la pregunta sean **verdaderas**. En los demás casos no mostramos el mensaje.

Con la conjunción **And** en el momento en el que aparece un **Falso** en alguna de las dos partes toda la expresión se considera falsa.

Or

La conjunción **Or** nos permite, al igual que **And** hacer una pregunta entre dos elementos. En este caso para que el resultado total sea **verdadero**, solo necesitamos que uno de los elementos sea verdadero.

Pongamos un ejemplo: imaginemos que tenemos una aplicación en la que queremos que se nos muestre un mensaje en el momento que una persona sea mayor de 25 años o esté casada. Vamos a escribir la línea de código correspondiente a este ejemplo:

If Edad.Text > 25 Or Casado.Value = True Then

Hagamos una tabla para ver en que casos se mostraría el mensaje .

Edad.Text	Casado.Value	Mensaje
Verdadero	Verdadero	Sí
Verdadero	Falso	Sí
Falso	Verdadero	Sí
Falso	Falso	No

Con la conjunción **Or** en el momento en el que aparece un **Verdadero**, toda la expresión se considera verdadera.

Not

Solo se utiliza para obtener la negación lógica de un objeto.

Un ejemplo de la utilización de **Not** lo vimos en la lección 2. Cuando al pulsar un botón queríamos poner o quitar la negrita, según si el texto a cambiar estaba o no en negrita.

La instrucción en cuestión era la siguiente: **Texto.FontBold = Not Texto.FontBold**. Si el valor que está detrás del **Not** es **verdadero** pasa a ser **falso** y si es **falso** pasa a ser **verdadero**.

Unión de las estructuras de decisión

En este apartado vamos a ver como podemos utilizar las estructuras de decisión en un mismo programa.

Según el caso que se nos plantee podremos ver que las estructuras de decisión las podemos utilizar una detrás de la otra o anidadas (una dentro de la otra).

Práctica 5

Vamos a realizar una práctica en la que utilizaremos las dos estructuras de decisión que hemos visto a lo largo de esta lección.

Crearemos una aplicación en la que después de seleccionar un elemento de entre una lista de objetos nos devuelva el precio. El precio podrá aparecer **con** o **sin IVA**.

Para crear esta aplicación nosotros solo indicaremos que objetos se necesitan y

alguna de las propiedades que se deberán cambiar. El aspecto de la aplicación no es lo más importante, pero puedes dedicarle un rato a perfeccionar la apariencia de los objetos insertados para hacer más atractiva la aplicación.

1. Inserta un **ListBox** al que llamarás **ListaObjetos**.
2. Introduce en el **ListBox** 5 nombres de diferentes objetos que puedas comprar en cualquier tienda a los que más tarde pondremos precio.
3. Inserta un **Frame**.
4. Dentro de este **Frame** inserta dos **OptionButton**.
5. El primero tendrá como nombre **ConIVA** y como **Caption: Con IVA**.
6. El segundo tendrá como nombre **SinIVA** y como **Caption: Sin IVA**.
7. Activa uno de los dos **OptionButton** que has insertado anteriormente.

Según el **OptionButton** que se seleccione se mostrará el precio **con** o **sin IVA**. Imaginemos que el IVA (Impuesto Valor Añadido) es de un 16% sobre el precio del producto.

8. Inserta un botón al que le llamaremos **MostrarPrecio** y como **Caption** tendrá **Mostrar Precio**.

Al pulsar dicho botón la aplicación nos enseñará el precio del producto señalado.

9. Inserta un **Label** al que le borraré el contenido y le pondremos como nombre **Precio**.

En dicho **Label** mostremos el precio del producto seleccionado.

Empezando a codificar

Para saber cual de los objetos de la lista está activado vamos a utilizar la propiedad **ListIndex** del objeto **ListaObjetos**, la cual nos devuelve el índice del elemento seleccionado. Recuerda que dicho índice siempre empieza a contar desde 0.

Vamos a utilizar la estructura de decisión **Select Case** la cual nos permitirá tomar un camino u otro en el código según el índice que esté seleccionado de nuestra lista.

Observa la estructura del código que más adelante escribiremos en el evento **Click** del botón **MostrarPrecio**.

```
Select Case ListaObjetos.ListIndex
    Case 0
        ` Instrucciones primer objeto seleccionado.
    Case 1
        ` Instrucciones segundo objeto seleccionado.
    Case 2
        ` Instrucciones tercer objeto seleccionado.
    Case 3
        ` Instrucciones cuarto objeto seleccionado.
    Case 4
        ` Instrucciones quinto objeto seleccionado.
End Select
```


Las líneas que tienen este símbolo al principio ' solo son líneas de comentario. El ordenador en el momento de ejecutar este código las pasará por alto. Es recomendable utilizar los comentarios para así facilitar la lectura del código.

El ordenado al pulsar el botón, miraría cual es el índice de la tabla que está seleccionado y ejecutaría las líneas de instrucciones que están dentro del índice indicado. En dichas líneas podemos hacer que el ordenador mire si tenemos seleccionado el precio **con** o **sin Iva** y nos lo muestre. También podríamos hacer que el precio de dicho objeto seleccionado lo guarde en una variable y después de la estructura **Select Case** mire si está seleccionado el Iva o no y se realicen los cálculos pertinentes en cada uno de los casos.

Vamos a poner los dos códigos y después miraremos cual de ellos es más correcto o más útil.

Primero escribiremos el código utilizando un **If** dentro de cada uno de los objetos seleccionados. Puedes poner los precios que tú quieras dependiendo de los objetos que hayas escrito dentro de la lista.

10. Completa el código de la siguiente manera.

```

Private Sub MostrarPrecio_Click()
  Select Case ListaObjetos.ListIndex
    Case 0
      If SinIVA.Value = True Then
        Precio.Caption = 1000
      Else
        Precio.Caption = (1000 * 16 / 100) + 1000
      End If
    Case 1
      If SinIVA.Value = True Then
        Precio.Caption = 2000
      Else
        Precio.Caption = (2000 * 16 / 100) + 2000
      End If
    Case 2
      If SinIVA.Value = True Then
        Precio.Caption = 3000
      Else
        Precio.Caption = (3000 * 16 / 100) + 3000
      End If
    Case 3
      If SinIVA.Value = True Then
        Precio.Caption = 4000
      Else
        Precio.Caption = (4000 * 16 / 100) + 4000
      End If
    Case 4
      If SinIVA.Value = True Then
        Precio.Caption = 5000
      Else
        Precio.Caption = (5000 * 16 / 100) + 5000
      End If
    End Select
  End Sub

```

Observa que el **IVA** lo calcula el ordenador, multiplicando el precio por un 16% (16 / 100) y sumándoselo al precio de dicho objeto.

11. Realiza un ejecución de prueba seleccionando diferentes objetos, indicando si deseas ver el precio **con** o **sin IVA**.

Ahora vamos a poner el mismo código utilizando la estructura **If** una sola vez y utilizando una variable.

```
Private Sub MostrarPrecio_Click()  
    Select Case ListaObjetos.ListIndex  
        Case 0  
            Precios = 1000  
        Case 1  
            Precios = 2000  
        Case 2  
            Precios = 3000  
        Case 3  
            Precios = 4000  
        Case 4  
            Precios = 5000  
    End Select  
    If SinIVA.Value = True Then  
        Precio.Caption = Precios  
    Else  
        Precio.Caption = Precios + (Precios * 16 / 100)  
    End If  
End Sub
```

Si te fijas para el primero de los dos códigos hemos utilizado muchas más líneas que para el segundo. Esto es debido a que en cada uno de los casos producidos por el índice de la lista poníamos una estructura **If**, para comprobar si teníamos seleccionado el precio **con** o **sin IVA**. En el segundo ejemplo solo ponemos la estructura **If** al final del código justo antes de mostrar el precio. En cada uno de los casos almacenamos el precio del objeto seleccionado en una variable para después poder calcular con él el precio total del objeto.

Tanto como en entendimiento, como en cantidad de líneas que ocupa el código es mucho mejor utilizar el segundo caso que no el primero. Siempre tenemos que evitar escribir líneas que puedan estar repetidas o que podamos evitarlas de alguna forma.

Fin lección 5